# Maintainer: Bruno Pagani <archange@archlinux.org>

pkgname=grammalecte
pkgver=2.1.1
pkgrel=3
pkgdesc="Grammar and typographic checker dedicated to the French language – standalone version"
arch=(any)
url="https://grammalecte.net/"
license=(GPL3)
depends=(python)
makedepends=(unzip python-setuptools)
options=(!emptydirs)
source=(https://grammalecte.net/grammalecte/zip/Grammalecte-fr-v${pkgver}.zip)
sha256sums=('f643eca603652469947d9bd85fb0ec08f8eb0f7b6ab925d2fb18bfc6fdd8d21c')

build() {
  python setup.py build
}

package() {
  python setup.py install --root="${pkgdir}" --skip-build --optimize=1
  mv "${pkgdir}"/usr/bin/grammalecte-cli{.py,}
  mv "${pkgdir}"/usr/bin/grammalecte-server{.py,}
}
